APPLICATION NOTE

PARAMETER DEPENDENT TOLERANCES - MATCHING MOS DEVICES

OVERVIEW

This note describes how to implement Monte Carlo tolerances that are dependent on individual device parameters such as the length and width of a MOSFET. Usually the matching tolerance between devices improves as their size increases and so a constant matching tolerance is inappropriate.

THE PROBLEM

Suppose we wish to run a Monte Carlo simulation to investigate the offset performance of a CMOS amplifier. To do this we need a tolerance specification for the matching of the Vt of the device. However this tolerance depends on the length and width (L and W) of the device and L and W are device parameters not model parameters. That is they are specified on the device line not in the .MODEL statement. But Vt is a model parameter, so to have a varying tolerance, we will need a separate .MODEL for each device instance.

SUBCIRCUIT IMPLEMENTATION

The problem described above isn't specific to implementing device dependent tolerances. The same problem would arise if you wanted to make any model parameter dependent on a device parameter.. As we shall see, however, Monte Carlo tolerances are a little more complicated.

The way to make a model parameter dependent on a device parameter is to place the device and its model inside a subcircuit. E.g.:

```
.SUBCKT N1 D G S B params: {W=1u} {L=1u}
M1 D G S B NN1 W={W} L={L}
.MODEL NN1 NMOS LEVEL=8 VTH0=0.5 .... etc
.ENDS
```

As the .MODEL statement is inside the subcircuit, a new one will be created for every subcircuit instance. This makes it possible to parameterise the model parameters with expressions dependent on the values passed to the subcircuit. In this case L and W.

MATCHING MOS DEVICES

According to Pelgrom *et al* (see reference) the matching of two MOS devices can be described by:

$$\sigma^2(V_{T0}) = \frac{A_{VT0}^2}{WL} + S_{VT0}^2 D^2$$

D is a measure of the distance between the devices. We can assume this to be 0 if the devices are adjacent to each other.. A_{VT0} is a process dependent parameter while W and L are width and length respectively.
A typical value for A_{VT0} might be 2E-8 but the actual value would need to be obtained from the process foundry.

Using the above equation, we arrive at the following subcircuit:

```
.PARAM AVTO=2e-8
.SUBCKT  N1 D G S B params: {W=1u} {L=1u}
M1 D G S B NN1 W={W} L={L}
.PARAM TOL_VT0={AVT0/sqrt(W*L)}
.MODEL NN1 NMOS LEVEL=8 VTH0={0.5*GAUSS(TOL_VT0*3)}
.ENDS
```

Note that we have multiplied the TOL_VT0 parameter by 3 because SIMetrix assumes 3-sigma tolerances.

LOT TOLERANCES

The above subcircuit only implements matching (or “DEV”) tolerances. We might also want to simulate the effect of “LOT” tolerances. A LOT tolerance is one that applies to all devices simulataneously. Usually in SIMetrix, LOT tolerances are implemented using the UNIFL or GAUSSL functions. These return the same value for each instance of a .MODEL. However, this approach cannot be used here because the each device uses its own model. To overcome this we must use an alternative technique that employs a global parameter. This is illustrated below:

```
.PARAM VT0_LOT_TOL=UNIF(0.2)
.PARAM AVTO=2e-8
.SUBCKT  N1 D G S B params: {W=1u} {L=1u}
M1 D G S B NN1 W={W} L={L}
.PARAM TOL_VT0={AVT0/sqrt(W*L)}
.MODEL NN1 NMOS LEVEL=8 VTH0={0.5*GAUSS(TOL_VT0*3)*VT0_LOT_TOL}
.ENDS
```

The parameter `VT0_LOT_TOL` has been assigned with a random value that will be calculated just once during each run. The same value will thus be used for every instance of the subcircuit.

REFERENCE

M. J. Pelgrom, A. C. J. Duinmaijer, and A. P. G. Welbers. "*Matching properties of MOS transistors*". IEEE J. SolidState Circuits, vol. 24, no. 5, pp. 1433--1439, Oct. 1989